
Handbook Organization

Welcome

This manual will introduce you to the HP Instrument BASIC programming language, provide
some helpful hints on getting the most use from it, and provide a general programming
reference. It is divided into three books, HP Instrument BASIC Programming Techniques, HP
Instrument BASIC Interfacing Techniques, and HP Instrument BASIC Language Reference.
The first two books provide some introductory material on programming and interfacing.
However, if you have no programming knowledge, you might find it helpful to study a
beginning-level programming book.

This manual assumes that you are familiar with the operation of HP Instrument BASIC’s
front-panel interface or keyboard and have read or reviewed the manual that describes the
operation of HP Instrument BASIC with your specific instrument.

HP Instrument BASIC is implemented as an “embedded controller”—that is, a computer
residing inside an instrument. Hence, all references in this manual to the “computer” also
refer to HP Instrument BASIC installed in an instrument.

What’s In This Handbook?

HP Instrument BASIC Programming Techniques contains explanations and programming
hints organized by concepts and topics. It is not a complete keyword reference. Instead it
covers programming concepts, showing how to use the HP Instrument BASIC language.

For explanations and hints regarding interfacing, see the HP Instrument BASIC Interfacing
Techniques book.

HP Instrument BASIC Language Reference contains a detailed keyword reference.

Handbook Organization 1



For HP BASIC Programmers

Many programmers already familiar with HP Series 200/300 BASIC will want to use the HP
Instrument BASIC manual set to look up keywords and find specifics about the way HP
Instrument BASIC is implemented. If this is your situation, you may want to refer to the
following instrument-specific manuals and sections as needed:

m The graphics section of your instrument-specific manual for information on using the display
for graphics and text program output.

m Your instrument-specific manual to learn how HP Instrument BASIC interfaces with the
host device, (if using an embedded controller) and its external GPIB port.

m Your instrument-specific manual for a description of how to transfer data between external
and internal programs, how to upload and download programs and how to control HP
Instrument BASIC programs from an external controller.

m “Keyword Guide to Porting” at the end of HP Instrument BASIC Programming Techniques
for a quick determination of what commands are implemented and how they relate to recent
versions of the corresponding HP Series 200/300 BASIC command.

Most importantly, you will find a complete command reference and a list of error messages in
the HP Instrument BASIC Language Reference. If you need to refresh your memory on any
other topics, consult the manuals on programming and interfacing techniques as needed.

Manual Organization

Welcome

This purpose of this manual is to introduce you to the HP Instrument BASIC programming
language and to provide some helpful hints on getting the most use from it. This manual
assumes that you are familiar with the operation of HP Instrument BASIC’s front-panel
interface or keyboard and have read or reviewed the manual that came with your instrument
that describes operation of HP Instrument BASIC with your specific instrument. Most topics
concerning running, recording, loading, saving and debugging programs are covered there.

This manual serves as a general language reference and programming tutorial for those with
a rudimentary knowledge of programming in BASIC or another language. If you have no
programming knowledge, you may find it helpful to study a beginning level programming
book. However, some beginners may find that they are able to start in this manual by
concentrating on the fundamentals presented in the first few chapters.

If you are a programming expert or are already familiar with the BASIC language of other HP
computers, you may start faster by going directly to the HP Instrument BASIC Language
Reference and checking the keywords you normally use.

HP Instrument BASIC is implemented as an “embedded controller”—that is, a computer
residing inside an instrument. Hence, all references in this manual to the “computer” also
refer to HP Instrument BASIC installed in an instrument.

What’s In This Manual

This manual contains explanations and programming hints organized by concepts and topics.
It is not an exhaustive keyword reference. Instead it covers programming concepts, showing
how to use the HP Instrument BASIC language. HP Instrument BASIC Language Reference
contains a detailed keyword reference. For explanations and hints regarding interfacing, see
the HP Instrument BASIC Interfacing Techniques book.

The following section gives an overview of the chapters in this manual.

Overview of Chapters

Chapter Topics

Chapter 2: Program This chapter describes program flow and how to control it.
Structure and Flow

Chapter 3: Numeric This chapter covers mathematical operations and the use of
Computation numeric variables.

Chapter 4: Numeric Arrays  This chapter covers numeric array operations.

Chapter 5: String
Manipulation

Chdpter 6: Subprograms and
User-Defined Functions

Chapter 7: Data Storage and
Retrieval

Chapter 8: Using a Printer
Chapter 9: Handling Errors

Chapter 10: Keyword Guide
to Porting

What’s Not in this Manual

This chapter explains the techniques used for the processing of
characters, words, and text in your program.

This chapter describes using alternate contexts (or
environments), available as user-defined functions or
subprograms.

This chapter shows many of the alternatives available for
storing the data that is intended as program input or created
as program output.

This chapter tells how to use an external printer, and how to
use formatted printing for both printer and CRT output.

This chapter discusses techniques for intercepting errors that
might occur while a program is running,.

This chapter summarizes the HP Instrument BASIC keywords
by categories, with differences between HP Instrument BASIC
and HP Series 200/300 BASIC.

This is a manual of programming techniques, helpful hints, and explanations of capabilities.
It is not a rigorous tutorial of the HP Instrument BASIC language. Any statements
appropriate to the topic being discussed are included in each chapter, whether they have
been previously introduced or not. Since most users will not read this manual from cover to
cover, the approach chosen should not present any significant problems. In cases where you
have difficulty getting the meaning of certain items from context, consult the Index to find

additional information.

Program Structure and Flow

There are four general categories of program flow. These are:
m Sequence

m Selection (conditional execution)

m Repetition

m Event-Initiated Branching

This chapter tells you how to use these types of program flow.

Sequence

The simplest form of sequence is linear flow. Linear flow allows many program lines
to be grouped together to perform a specific task in a predictable manner. Keep these
characteristics of linear flow in mind:

m Linear flow involves no decision making. Unless there is an error condition, the program
lines will always be executed in exactly the same order.

m Linear flow is the default mode of program execution. Unless you include a statement that
stops or alters program flow, the computer will always execute the next higher-numbered
line after finishing the line it is on.

Halting Program Execution
There are three statements that can halt program flow: END, STOP, and PAUSE.

The END Statement

The primary purpose of the END statement is to mark the end of the main program. When
an END statement is executed, program flow stops and the program moves into the stopped
(non-continuable) state.

The STOP Statement

The STOP statement acts like an END statement in that it stops program flow. You can use
a STOP statement to halt program flow at some point other than the end of the program.
When a STOP statement is executed, program flow stops and the program moves into the
stopped (non-continuable) state.

Program Structure and Flow 2-1



The PAUSE Statement

Use the PAUSE statement to temporarily halt program execution, leaving the program
variables intact. Execution halts until instructed to continue by the operator.

Following is an example of the use of PAUSE:

100 Radius=5

110 Circum=PI*2*Radius
120 PRINT INT(Circum)
130 PAUSE

140 Area=PI*Radius~2
150 PRINT INT(Area)
160 END

When the program runs, the computer prints 31 on the CRT. Then when you continue, the
computer prints 78 on the CRT. One common use for the PAUSE statement is in program
troubleshooting and debugging. Another use for PAUSE is to allow time for the computer
user to read messages or follow instructions.

Simple Branching

An alternative to linear flow is branching. Branching is simply a redirection of sequential flow.
The simplest form of branching uses the statements GOTO and GOSUB. Both statements
cause an unconditional branch to a specified location in a program.

Using GOTO

The GOTO statement causes the program to branch to either a line number or the line label.
Following are examples of the GOTO statement:

30 REM GOTO branches here
100 GOTO 30
150 GOTO Label_xyz

300 Label_xyz:...

Using GOSUB

The GOSUB statement transfers program execution to a subroutine. A subroutine is simply a
segment of a program that is entered with a GOSUB and exited with a RETURN. There are
no parameters passed and no local variables are allowed in the subroutine.

The GOSUB is very useful in structuring and controlling programs. It is similar to a
procedure call in that program flow automatically returns to the line following the GOSUB
statement. The GOSUB statement can specify either the line label or the line number of the
desired subroutine entry point. The following are examples of GOSUB statements:

100 GOSUB 1000
200 GOSUB Label_abc

1000 REM subroutine begins here
1010 Label_abc:

1500 RETURN

Remember that each time a subroutine is called by a GOSUB, control returns to the line
immediately following the GOSUB when the RETURN is encountered in the subroutine. Note
that if you omit the RETURN statement in a subroutine the program will continue executing
beyond the point at which you expected it to return, until it encounters another RETURN or
one of the halting statements (PAUSE, STOP, or END).

Selection

The heart of a computer’s decision-making power is the category of program flow called
selection, or conditional execution. As the name implies, a certain segment of the program
either is or is not executed according to the results of a test or condition. This section
presents the conditional-execution statements according to various applications. The following
is a summary of these groupings.

m Conditional execution of one segment.
m Conditionally choosing one of two segments.

m Conditionally choosing one of many segments.

Conditional Execution of One Segment

The basic decision to execute or not execute a program segment is made by the IF ... THEN

statement. This statement includes a numeric expression that is evaluated as being either true
or false. If true (non-zero), the conditional segment is executed. If false (zero), the conditional
segment is bypassed. Note that any valid numeric expression is allowed for the test expression.

The conditional segment can be either a single HP Instrument BASIC statement or a program
segment containing any number of statements. The first example shows conditional execution
of a single statement.

100 IF Ph>7.7 THEN PRINT "Ph Value has been exceeded!"

Notice the test (Ph>7.7) and the conditional statement (Print “Ph Value ... ”) that appear
on either side of the keyword THEN. When the computer executes this program line, it
evaluates the expression Ph>7.7. If the value contained in the variable Ph is 7.7 or less, the
expression evaluates to 0 (false), and the line is exited. If the value contained in the variable
Ph is greater than 7.7, the expression evaluates as 1 (true), and the PRINT statement is
executed.

Prqhibited Statements

Certain statements are not allowed as the conditional statement in a single-line IF ... THEN.
The following statements are not allowed in a single-line IF ... THEN.

Keywords used in the declaration of variables:

COM DIM INTEGER REAL

Keywords that define context boundaries:

DEF FN FNEND SUB SUBEND END

Keywords that define program structures:

CASE END LOOP FOR REPEAT
CASE ELSE END SELECT IF SELECT
ELSE END WHILE LOOP UNTIL
END IF EXIT IF NEXT WHILE

Keywords used to identify lines that are literals:

DATA REM

Conditional Branching

Powerful control structures can be developed by using branching statements in an
IF ... THEN. For example:

110 TIF Free_space<100 THEN GOSUB Expand_file
120 ! The line after is always executed

This statement checks the value of a variable called Free_space, and executes a file-expansion
subroutine if the value tested is not large enough. One important feature of this structure

is that the program flow is essentially linear, except for the conditional “side trip” to a
subroutine and back.

The conditional GOTO is such a commonly used technique that the computer allows a special
case of syntax to specify it. Assuming that line number 200 is labeled “Start”, the following
statements will all cause a branch to line 200 if X is equal to 3.

IF X=3 THEN GOTO 200
IF X=3 THEN GOTO Start
IF X=3 THEN 200

IF X=3 THEN Start

When a line number or line label is specified immediately after THEN, the computer assumes
a GOTO statement for that line. This improves the readability of programs.
Multiple-Line Conditional Segments

If the conditional program segment requires more than one statement, a slightly different
structure is used. For example:

100 IF Ph>7.7 THEN

110 PRINT "The value of Ph has been exceeded!"
120 PRINT "Final Ph =";Ph

130 GOSUB Next_tube

140 END IF

150 ! Program continues here

If Ph is less than or equal to 7.7 the program skips all of the statements between the
IF..THEN and the END IF statements and continues with the line following the END IF
statement. If Ph is greater than 7.7, the computer executes the three statements between the
IF ... THEN and END IF statements. Program flow then continues at line 150. Any number
of program lines can be placed between a THEN and an END IF statement including other
IF..END IF statements. Including other IF..END IF statements is called nesting or nested
constructs. For example:

1000 IF Flag THEN
1010 IF End_of_page THEN

1020 FOR I=1 TO Skip_length
1030 PRINT

1040 Lines=Lines+1

1050 NEXT I

1060 END IF

1070 END IF

Choosing One of Two Segments

Often you want a program flow that passes through only one of two paths depending upon a
condition. This type of decision is shown in the following diagram:

Flag = 1 Flag = 0
I
| 400 IF Flag THEN —-
|
I

410 R=R+2 I

420 Area=PI*R"2 |

--- 430 ELSE <=
I 440 Width=Width+1 |
| 450 Length=Length+1 I
| 460 Area=Width*Length |
I 470 END IF !
--> 480 Print "Area ='";Area |
| 490 ! Program continues |
v v

HP Instrument BASIC has an IF ... THEN ... ELSE structure that makes the one-of-two
choice easy and readable.

Choosing One of Many Segments

The SELECT ... END SELECT is similar to the IF ... THEN ... ELSE ... END IF
construct, but allows the definition of several conditional program segments. Only one
segment executes each time the construct is entered. Each segment starts after a CASE or
CASE ELSE statement, and ends when the next program line is a CASE, CASE ELSE, or
END SELECT statement.

Consider for example, the processing of readings from a voltmeter. Readings have been
entered that contain a function code. These function codes identify the type of reading and
are shown in the following table:

Function Code Type of Reading
DV DC Volts
AV AC Volts
DI DC Current
Al AC Current
oM Ohms

This example shows the use of the SELECT construct. The function code is contained in the
variable Funct$. The rules about illegal statements and proper nesting are the same as those
for the IF ... THEN statement.

2000 SELECT Funct$
2010 CASE "DV"

2020 !

2030 ! Processing for DC Volts
2040 !

2050 CASE "AV"

2060 !

2070 ! Processing for AC Volts
2080 !

2090 CASE "DI"

2100 !

2110 ! Processing for DC Amps
2120 !

2130 CASE "AI"

2140 !

2150 ! Processing for AC Amps
2160 !

2170 CASE "OM"

2180 !

2190 ! Processing for Ohms
2200 !

2210 CASE ELSE

2220 BEEP

2230 PRINT "INVALID READING"
2240 END SELECT
2250 ! Program execution continues here

Notice that the SELECT construct starts with a SELECT statement specifying the variable
to be tested and ends with an END SELECT statement. The anticipated values are placed
in CASE statements. Although this example shows a string tested against simple literals,
the SELECT statement works for numeric or string variables or expressions. The CASE
statements can contain constants, variables, expressions, comparison operators, or a range
specification. The anticipated values, or match items, must be of the same type (numeric or
string) as the tested variable.

The CASE ELSE statement is optional. 1t defines a program segment that is executed if the
tested variable does not match any of the cases. If CASE ELSE is not included and no match
is found, program execution simply continues with the line following END SELECT.

A CASE statement can also specify multiple matches by separating them with commas, as
follows:

CASE -1,1,3T0 7,>15

If an error occurs when the computer tries to evaluate an expression in a CASE statement,
the error is reported for the line containing the SELECT statement. An error message
pointing to a SELECT statement actually means that there was an error in that line or in one
of the CASE statements following it.

Repetition

There are four structures available for creating repetition. The FOR ... NEXT structure
repeats a program segment a predetermined number of times. Two other structures,
REPEAT ... UNTIL and WHILE ... END WHILE, repeat a program segment indefinitely,
waiting for a specified condition to occur. The LOOP ... EXIT IF structure is used to create
an iterative structure that allows multiple exit points at arbitrary locations.

Fixed Number of Iterations

The general concept of repetitive program flow can be shown with the FOR ... NEXT
structure. The FOR statement marks the beginning of the repeated segment and establishes
the number of repetitions. The NEXT statement marks the end of the repeated segment.
This structure uses a numeric variable as a loop counter. This variable is available for use
within the loop, if desired. The following example shows the basic elements of a FOR ...
NEXT loop.

10  FOR X=10 TO 0 STEP -1

20 BEEP
30 PRINT X
40 WAIT 1
50 NEXT X
60 END

In this example, X is the loop counter, 10 is the starting value, 0 is the final value, -1 is the
step size and the repeated segment is composed of lines 20 through 50. Note that if the step
counter is not specified, a default value of 1 is assumed.

When all variables involved are integers, the number of iterations can be predicted using the
following formula:

INT((Step_Size + Final_Value — Starting_Value)/(Step_Size))

Thus, the number of iterations in the example above is 11.

Conditional Number of Iterations

Some applications need a loop that is executed until a certain condition is true regardless of
the number of loop iterations required. The REPEAT ... UNTIL and the WHILE ... END

WHILE structures provide this flexibility.

The REPEAT loop and the WHILE loop differ only in their location of the loop exit test.
The REPEAT loop has its test at the end of the loop. Therefore, the loop will always be
executed once because the condition is not tested until the end of the loop. The WHILE
loop has its test at the beginning of the loop, so the test is made before the loop is entered.
Therefore, it is possible for a WHILE loop to be skipped entirely.

For example, suppose you wat to print successive powers of two, but want to stop once the
value is greater than 1000. Consider the following examples programs:

REPEAT loop

10 X=2

20 PRINT X;

30 REPEAT

40  X=X*2

50  PRINT X;
60 UNTIL X>1000
70 END

WHILE loop

10 X=2

20 PRINT X;

30 WHILE X<1000
40  X=X*2

50  PRINT X;
60 END WHILE

70 END

If you ran either of these programs, the results would be:
2 4 8 16 32 64 256 512 1024
However, if you replace line 10 in each program with
10 X=1024
then the repeat loop would produce
1024 2048
whereas the WHILE loop would produce
1024

Arbitrary Exit Points

The looping structures discussed so far allow only one exit point. There are times when this is
not the desired program flow. The LOOP..EXIT IF construct allows you to have any number
of conditional exits points. Also, the EXIT IF statement can be at the top or bottom of the
loop. This means that the LOOP structure can serve the same purposes as REPEAT ...
UNTIL and WHILE ... END WHILE.

The EXIT IF statement must appear at the same nesting level as the LOOP statement for a
given loop. This is best shown with an example. In the “WRONG” example, the EXIT IF

statement has been nested one level deeper than the LOOP statement because it was placed
in an IF ... THEN structure.

WRONG:

600 LOOP
610 Test=RND-.5
620 IF Test<0 THEN

630 GOSUB Negative
640 ELSE

650 EXIT IF Test>.4
660 GOSUB Positive

670 END IF
680 END LOOP

RIGHT:

Here is the proper structure to use.

600 LOOP

610 Test=RND-.5
620 EXIT IF Test>.4
630 IF Test<0 THEN

640 GOSUB Negative
650 ELSE
660 GOSUB Positive

670 END IF
680 END LOOP

Event-Initiated Branching

HP Instrument BASIC provides a tool called event-initiated branching that uses interrupts

to redirect program flow. Each time the program finishes a line, the computer executes an
“event-checking” routine. If an enabled event has occurred, then this “event-checking” routine
causes the program to branch to a specified statement.

Types of Events

Event-initiated branching is established by the ON..event statements. Here is a list of the
statements:

ON ERROR an interrupt generated by a run-time error
ON INTR an interrupt generated by an an interface
ON KEY an interrupt generated by pressing a softkey

ON TIMEOUT an interrupt generated when an interface or device has taken longer than a
specified time to respond to a data-transfer handshake

The following example demonstrates an event-initiated branch using the ON KEY statement.

100 ON KEY 1 LABEL "Inc'" GOSUB Plus
110 ON KEY 5 LABEL "Dec" GOSUB Minus
120 ON KEY 8 LABEL "Abort'" GOTO Bye

130 !

140  Spin: DISP X
150 GOTO Spin
160 !

170 Plus: X=X+1
180 RETURN

190 !
200 Minus: X=X-1
210 RETURN

220 Bye: END

The ON KEY statements are executed only once at the start of the program. Once defined,
these event-initiated branches remain in effect for the rest of the program.

The program segment labeled “Spin” is an infinite loop. If it weren’t for interrupts,

this program couldn’t do anything except display a zero. However, there is an implied

“IF ... THEN” at the end of each program line due to the ON KEY action. As a result of
softkey presses, either the “Plus” or the “Minus” subroutines are selected or the program
branches to the END statement and terminates. If no softkey is pressed, the computer
continues to display the value of X.

The following section of “pseudo-code” shows what the program flow of the “Spin” segment
actually looks like to the computer.
Spin: display X
if Keyl then gosub Plus
if Key5 then gosub Minus
if Key9 then goto Bye
goto Spin

The labels are arranged to correspond to the layout of the softkeys. The labels are displayed
when the softkeys are active and are not displayed when the softkeys are not active. Any

label that your program has not defined is blank. The label areas are defined in the ON KEY
statement by using the keyword LABEL followed by a string.

Deactivating Events

All the “ON-event” statements have a corresponding “OFF-event” statement. This is one
way to deactivate an interrupt source. For example OFF KEY deactivates interrupts from the
softkeys. Pressing a softkey while deactivated does nothing.

Disabling Events

It is also possible to temporarily disable an event-initiated branch. This is done when an
active event is desired in a process, but there is a special section of the program that you
don’t want to be interrupted. Since it is impossible to predict when an external event will
occur, the special section of code can be “protected” with a DISABLE statement.

100 ON KEY 9 LABEL " ABORT" GOTO Leave
110 !

120 Print_line: !

130 DISABLE

140 FOR I=1 TO 10

150 PRINT I;

160 WAIT .3

170 NEXT I

180  PRINT

190 ENABLE

200 GOTO Print_line
210 !

220 Leave: END

This example shows a DISABLE and ENABLE statement used to “frame” the Print_line
segment of the program. The “ABORT” key is active during the entire program, but the
branch to exit the routine will not be taken until an entire line is printed. The operator can
press the “ABORT” key at any time. The key press will be logged, or remembered, by the
computer. Then when the ENABLE statement is executed, the event-initiated branch is
taken.

Chaining Programs

With HP Instrument BASIC, it is also possible to “chain” programs together; that is, one
program may be executed, which, in turn, loads and runs another. This method is often used
when you have several large program segments that will not all fit into memory at the same
time. This section describes program chaining methods.

Using GET

The GET command brings in programs or program segments from an ASCII file, with the
options of appending them to an existing program and/or beginning program execution at a
specified line.

The following statement:
GET "George",100

first deletes all program lines from 100 to the end of the program, then appends the lines in
the file named “George” to the lines that remained at the beginning of the program. The
program lines in file “George” would be renumbered to start with line 100.

GET can also specify where program execution begins. This is done by specifying two line
identifiers. For example:

100 GET "RATES",Append_line,Run_line
specifies that:

1. Existing program lines from the line label “Append_line” to the end of the program are to
be deleted.

2. Program lines in the file named “RATES” are to be appended to the current program,
beginning at the line labeled “Append_line”; lines of “RATES” are renumbered if
necessary.

3. Program execution is to resume at the line labeled “Run_line”.

Although any combination of line identifiers is allowed, the line specified as the start of
execution must be in the main program segment (not in a SUB or user-defined function).
Execution will not begin if there was an error during the GET operation.

Example of Chaining with GET

A large program can be divided into smaller segments that are run separately by using GET.
The following example shows a technique for implementing this method.

First Program Segment:

10 COM Ohms,Amps,Volts
20 Ohms=120

30 Volts=240

40 Amps=Volts/Ohms

50 GET "Wattage"

60 END

Program Segment in File Named “Wattage”:

10 COM Ohms,Amps,Volts

20 Watts=Amps*Volts

30 PRINT "Resistance (in ohms)
40 PRINT "Power usage (in watts)
50 END

'; Ohms
";Watts

Lines 10 through 40 of the first program are executed in normal, serial fashion. Upon reaching
line 50, the system deletes all program lines of the program, then GETs the lines of the
“Wattage” program. Note that since they have similar COM declarations, the COM variables
are preserved and used by the second program. This feature is very handy to have while
chaining programs.

Program-to-Program Communications

As shown in the preceding example, if chained programs are to communicate with one
another, you can place values to be communicated in COM variables. The only restriction
is that these COM declarations must match ezxactly, or the existing COM will be cleared
when the chained program is loaded. For a description of using COM declarations, see the
“Subprograms” chapter of this manual.

One important point to note is the use of the COM statement. The COM statement places
variables in a section of memory that is preserved during the GET operation. Since the
program saved in the file named “Wattage” also has a COM statement that contains three
scalar REAL variables, the COM is preserved (it matches the COM declaration of the
“Wattage” program being appended with GET).

If the program segments did not contain matching COM declarations, all variables in the
mismatched COM statements would be destroyed by the “pre-run” that the system performs
after appending the new lines but before running the first program line.

Numeric Computation

Numeric computations deal exclusively with numeric values. Adding two numbers and finding
a sine or a logarithm are all numeric operations, but converting bases and converting a
number to a string or a string to a number are not.

Numeric Data Types

There are two numeric data types available in HP Instrument BASIC: INTEGER, and REAL.
Any numeric variable not declared INTEGER is a REAL variable. This section covers these
two numeric data types.

INTEGER Data Type
An INTEGER variable can have any whole-number value from —32 768 through +32 767.

REAL Data Type

A REAL variable can be any value from :—1.797 693 134 862 315 x 103°® through
1.797 693 134 862 315 x 1038, The smallest non-zero REAL value allowed is approximately
+ 2.225 073 858 507 202 x 10398,

A REAL can also have the value of zero.

REAL and INTEGER variables may be declared as arrays.

Declaring Variables

You can declare variables to be of a particular type by using the INTEGER and REAL
statements. For example, the statements:

INTEGER I, J, Days(5), Weeks(5:17)
REAL X, Y, Voltage(4), Hours(5,8:13)

each declare two scalar and two array variables. A scalar variable represents a single value.
An array is a subscripted variable that contains multiple values accessed by subscripts. You
can specify both the lower and upper bounds of an array or specify the upper bound only, and
use the default lower bound of 0. You can also declare an array using the DIM statement.

DIM R(4,5)

Assigning Variables

The most fundamental numeric operation is the assignment operation, achieved with the LET
statement. The LET statement may be used with or without the keyword LET. Thus, the
following statements are equivalent:

LET A=A+ 1
A=A+1

Implicit Type Conversions

The computer will automatically convert between REAL and INTEGER values in assignment
statements and when parameters are passed by value in function and subprogram calls. When
a value is assigned to a variable, the value is converted to the data type of that variable.

For example, the following program shows a REAL value being converted to an INTEGER:

100 REAL Real_var

110 INTEGER Integer_var

120 Real_var = 2.34

130 Integer_var = Real_var ! Type conversion occurs here.
140 DISP Real_var, Integer_var

150 END

Executing this program returns the following result:
2.34 2

When parameters are passed by value, the type conversion is from the data type of the calling
statement’s parameter to the data type of the subprogram’s parameter. When parameters are
passed by reference, the type conversion is not made and a TYPE MISMATCH error will be
reported if the calling parameter and the subprogram parameters are different types.

When a REAL number is converted to an INTEGER, the fractional part is lost and the
REAL number is rounded to the closest INTEGER value. Converting the number back to a
REAL will not restore the fractional part. Also, because of the differences in ranges between
these two data types, not all REAL values can be rounded into an equivalent INTEGER
value. This problem can generate INTEGER OVERFLOW errors.

The rounding problem does not generate an execution error. The range problem can generate
an execution error, and you should protect yourself from this possibility.

The following program segment shows a method to protect against INTEGER overflow errors
(note that the variable X is REAL and Y is INTEGER):

200 IF (-32768<=X) AND (X<=32767) THEN

210 Y=X

220 ELSE

230 GOSUB Out_of_range
240 END IF

It is possible to achieve the same effect using MAX and MIN functions:
200 Y=MAX(MIN(x,32767),-32768)

Both these methods avoid the overflow errors, but only the first includes the fact that
the values were originally out of range. If out-of-range is a meaningful condition, an error
handling trap is more appropriate.

Evaluating Scalar Expressions
This section covers the following topics as they relate to evaluating scalar expressions.
m Hierarchy of expression evaluation

m HP Instrument BASIC operators: monadic, dyadic, and relational

The Hierarchy

If you look at the expression 24+4/2+6, it can be interpreted several ways:
m 2+(4/2)46 = 10

m (2+4)/246 =9

m 2+4/(246) = 2.5

m (2+4)/(246) = .75

To eliminate this ambiguity HP Instrument BASIC uses a hierarchy for evaluating expressions.
In order to understand how HP Instrument BASIC evaluates these expressions, let’s examine
the valid elements in an expression and the evaluation hierarchy (the order of evaluation of the
elements).

Six items can appear in a numeric expression:

m Operators (+, —, etc.)—modify other elements of the expression.

m Constants (7.5, 10, etc.)—represent literal, non-changing numeric values.
m Variables (Amount, X_coord, etc.)—represent changeable numeric values.

m Intrinsic functions (SQRT, DIV, etc.)—return a value that replaces them in the evaluation
of the expression.

m User-defined functions (FNMy_func, FNReturn_val, etc.)—also return a value that replaces
them in the evaluation of the expression.

m Parentheses—are used to modify the evaluation hierarchy.

The following table defines the hierarchy used by the computer in evaluating numeric
expressions.

Math Hierarchy

Precedence Operator

Highest Parentheses; they may be used to force any order of operation
Functions, both user-defined and intrinsic

Exponentiation: "

Multiplication and division: * / MOD DIV MODULO
Addition, subtraction, monadic plus and minus: + —
Relational Operators: = <> < > <= >=

NOT

AND

Lowest OR, EXOR

When an expression is being evaluated it is read from left to right and operations are
performed as encountered, unless a higher precedence operation is found immediately to
the right of the operation encountered, or unless the hierarchy is modified by parentheses.
If HP Instrument BASIC cannot deal immediately with the operation, it is stacked, and the
evaluator continues to read until it encounters an operation it can perform. It is easier to
understand if you see an example of how an expression is actually evaluated.

The following expression is complex enough to demonstrate most of what goes on in
expression evaluation.

A = 5+3%(4+2) /SIN(X) +X* (1>X) +FNNeg1* (X<5 AND X>0)

To evaluate this expression, it is necessary to have some historical data. We will assume that
DEG has been executed earlier, that X= 90, and that FNNeg1 returns -1. Evaluation proceeds
as follows:

5+3%(4+2) /SIN(X)+X*(1>X) +FNNeg1*(X<5 AND X>0)
5+3%6/SIN(X) +X* (1>X) +FNNeg1*(X<5 AND X>0)
5+18/SIN(X)+X*(1>X)+FNNeg1*(X<5 AND X>0)
5+18/1+X*(1>X)+FNNeg1#(X<5 AND X>0)
5+18+X* (1>X) +FNNeg1* (X<5 AND X>0)

23+X* (1>X) +FNNeg1# (X<5 AND X>0)
23+X*0+FNNeg1*(X<5 AND X>0)
23+0+FNNeg1#(X<5 AND X>0)

23+FNNeg1*(X<5 AND X>0)

23+-1%(X<5 AND X>0)

23+-1%(0 AND X>0)

23+-1%(0 AND 1)

23+-1%0

23+0

23

Operators
There are three types of operators in HP Instrument BASIC: monadic, dyadic, and relational.

m A monadic operator performs its operation on the expression immediately to its right. + -
NOT

m A dyadic operator performs its operation on the two values it is between. The operators are
as follows: ~, *, /, MOD, DIV, +, -, =, <>, <, >, <=, >=, AND, OR, and EXOR.

m A relational operator returns a 1 (true) or a 0 (false) based on the result of a relational test
of the operands it separates. The relational operators are a subset of the dyadic operators
that are considered to produce Boolean results. These operators are as follows: <, >, <=,
>=, =, and <>.

While the use of most operators is obvious from the descriptions in the language reference,
some of the operators have uses and side effects that are not always apparent.

Expressions as Pass Parameters

All numeric expressions are passed by value to subprograms. Thus, 5+X is obviously passed
by value. Not quite so obviously, +X is also passed by value. The monadic operator makes it
an expression.

For more information on pass parameters, read the chapter entitled “Subprograms and
User-Defined Functions.”

Strings in Numeric Expressions

String expressions can be directly included in numeric expressions if they are separated by
relational operators. The relational operators always yield Boolean results, and Boolean
results are numeric values in HP Instrument BASIC. For example:

110 Day_number=1*(Day$="Sun")+2*(Day$="Mon")

Executing the program line above would result in Day_number being equal to 1 if Day$ equals
“Sun” and 2 if Day$ equals "Mon" (or 0 otherwise).

Step Functions

The comparison operators are useful for conditional branching (IF ... THEN statements), but
are also valuable for creating numeric expressions representing step functions. For example,
suppose you want to output certain values depending on the value, or range of values of a
single variable. This is shown as follows:

m If variable < 0 then output =0
m If 0 < variable < 1 then output equals the square root of (A% + B2).
m If variable > 1 then output = 15

It is possible to generate the required response through a series of IF ... THEN statements,
but it can also be done with the following expression (where X is the variable and Y is the
output):

Y=(X<0)*0+(X>=0 AND X<1)#* SQR(A"2+B"2)+(X>=1)*15

The Boolean expressions each return a 1 or 0, which is then multiplied by the accompanying
expression. Expressions not matching the selection return 0, and are not included in the
result. The value assigned to the variable (X) before the expression is evaluated determines
the computation placed in the result.

Comparing REAL Numbers

When you compare INTEGER numbers, no special precautions are necessary since these
values are represented exactly. However, when you compare REAL numbers, especially those
that are the results of calculations and functions, it is possible to run into problems due to
rounding. For example, consider the use of comparison operators in IF ... THEN statements
to check for equality in any situation resembling the following:

100  DEG

110 A=25.3765477
120  IF SIN(A)“"2+C0S(A)"2=1.0 THEN

130 PRINT "Equal"

140 ELSE

150 PRINT "Not Equal"
160 END IF

You will find that the equality test fails due to rounding errors. Irrational numbers and most
repeating decimals cannot be represented exactly in any finite machine, and most rational
decimal numbers with fractional parts cannot be represented exactly with binary numbers,
which HP Instrument BASIC uses internally.

Resident Numerical Functions

The resident functions are the functions that are part of the HP Instrument BASIC language.
Numerous functions are included to make mathematical operations easier. This section covers
these functions by placing them in the categories given below.

m Arithmetic Functions

® Exponential Functions

m Trigonometric Functions

m Binary Functions

m Limit Functions

m Rounding Functions

m Random Number Function

m Base Conversion Functions

m General Functions

Arithmetic Functions

HP Instrument BASIC provides you with the following functions:

ABS

FRACT

INT

MAXREAL

MINREAL

SQRT or SQR

SGN

Returns the absolute value of an expression. Takes a REAL, or INTEGER
number as its argument.

Returns the “fractional” part of the argument.

Returns the greatest integer that is less than or equal to an expression.
The result is of the same type (INTEGER or REAL) as the original
number.

Returns the largest positive REAL number available in HP Instrument
BASIC. Its value is approximately 1.797 693 134 862 32E+308.

Returns the smallest positive REAL number available in HP Instrument
BASIC. Its value is approximately 2.225 073 858 507 24E—308.

Return the square root of an expression. Takes a REAL or INTEGER
number as their argument.

Returns the sign of an expression: 1 if positive, 0 if 0, —1 if negative.

Exponential Functions

These functions determine the natural and common logarithm of an expression, as well as the
Napierian e raised to the power of an expression. Note that all exponential functions take
REAL, or INTEGER numbers as their argument.

EXP
LGT
LOG

Raise the Napierian e to an power. e = 2.718 281 828 459 05.
Returns the base 10 logarithm of an expression.

Returns the natural logarithm (Napierian base e) of an expression.

Trigonometric Functions

Six trigonometric functions and the constant m are provided for dealing with angles and
angular measure. Note that all trigonometric functions take REAL or INTEGER numbers as
their argument.

ACS Returns the arccosine of an expression.

ASN Returns the arcsine of an expression.

ATN Returns the arctangent of an expression.

cos Returns the cosine of the angle represented by the expression.

SIN Returns the sine of the angle represented by an expression.

TAN Returns the tangent of the angle represented by an expression.

PI Returns the constant 3.141 592 653 589 79, an approximate value for pi.

Trigonometric Modes: Degrees and Radians

The default mode for all angular measure is radians. Degrees can be selected with the DEG
statement. Radians may be reselected by the RAD statement. It is a good idea to explicitly
set a mode for any angular calculations, even if you are using the default (radian) mode. This
is especially important in writing subprograms, as the subprogram inherits the angular mode
from the context that calls it. The angle mode is part of the calling context.

Binary Functions

All operations that HP Instrument BASIC performs use a binary number representation. You
usually don’t see this, because HP Instrument BASIC changes decimal numbers you input
into its own binary representation, performs operations using these binary numbers, and then
changes them back to their decimal representation before displaying or printing them.

The following HP Instrument BASIC functions deal with binary numbers:

BINAND Returns the bit-by-bit “logical and” of two arguments.

BINCMP Returns the bit-by-bit “complement” of its argument.

BINEGOR Returns the bit-by-bit “exclusive or” of two arguments.

BINIOR Returns the bit-by-bit “inclusive or” of two arguments.

BIT Returns the state of a specified bit of the argument.

ROTATE Returns a value obtained by shifting an INTEGER representation of an
argument a specific number of bit positions, with wraparound.

SHIFT Returns a value obtained by shifting an INTEGER representation of an

argument a specific number of bit positions, without wraparound.

When any of these functions are used, the arguments are first converted to INTEGER (if they
are not already in the correct form), then the specified operation is performed. It is best to
restrict bit-oriented binary operations to be declared INTEGERs. If it is necessary to operate
on a REAL, make sure the precautions described under “Conversions,” at the beginning of
this chapter, are employed to avoid INTEGER overflow.

Limit Functions

It is sometimes necessary to find the range of values in a list of variables. HP Instrument
BASIC provides two functions for this purpose:

MAX Returns a value equal to the greatest value in the list of arguments.

MIN Returns a value equal to the least value in the list of arguments.

Rounding Functions

Sometimes it is necessary to round a number in a calculation to eliminate unwanted
resolution. There are two basic types of rounding, rounding to a total number of decimal
digits and rounding to a number of decimal places (limiting fractional information). Both
types of rounding have their own application in programming.

The functions that perform the types of rounding mentioned above are as follows:

DROUND Rounds a numeric expression to the specified number of digits. If the
specified number of digits is greater than 15, no rounding takes place. If
the number of digits specified is less than 1, zero is returned.

PROUND Returns the value of the argument rounded to a specified power of ten.

Random Number Function

The RND function returns a pseudo-random number between 0 and 1. Since many
applications require random numbers with arbitrary ranges, it is necessary to scale the
numbers.

200 R= INT(RND*Range)+0ffset
The above statement will return an integer between Offset and Offset + Range.

The random number generator is seeded with the value 37 480 660 at power-on, SCRATCH,
SCRATCH A, and pre-run. The pattern period is 23! — 2. You can change the seed with the
RANDOMIZE statement, which will give a new pattern of numbers.

Time and Date Functions
The following functions return the time and date in seconds:
TIMEDATE  Returns the current clock value (in Julian seconds).
For example, the statement
TIMEDATE
returns a value in seconds similar to the following:

2.11404868285E+11

Base Conversion Functions

The two functions IVAL and DVAL convert a binary, octal, decimal, or hexadecimal string
value into a decimal number.

IVAL returns the INTEGER decimal value of a binary, octal, decimal, or hexadecimal
16-bit integer. The first argument is a string and the second argument is the radix
or base to convert from. For example,

IVAL("12740",8)
returns the following numeric value
5600

DVAL returns the decimal whole number value of a binary, octal, decimal, or hexadecimal
32-bit integer. The first argument is a string and the second argument is the radix
or base to convert from. For example,

DVAL("11111111111111111114111111111100",2)
returns the following numeric value:
-4

For more information and examples of these functions, read the section “Number-Base
Conversion” found in the “String Manipulation” chapter.

General Functions

When you are specifying select code and device selector numbers, it is more descriptive to use
a function to represent that device as opposed to a numeric value. For example, the following
command allows you to enter a numeric value from the keyboard.

ENTER 2;Numeric_value

The above statement used in a program is not as easy to read as this one is:
ENTER KBD ;Numeric_value

where you know the function KBD must stand for keyboard.

Functions that return a select code or device selector are listed below:

CRT Returns the INTEGER 1. This is the select code of the internal CRT.
KBD Returns the INTEGER 2. This is the select code of the keyboard.
PRT Returns the INTEGER 701.

4

Numeric Arrays

An array is a multi-dimensioned structure of variables that are given a common name. The
array can have one to six dimensions. Each location in an array contains one value, and each
value has the characteristics of a single variable, either REAL or INTEGER (string arrays are
discussed in the chapter, “String Manipulation”).

A one-dimensional array consists of n elements, each identified by a single subscript. A
two-dimensional array consists of m times n elements where m and n are the maximum
number of elements in the two respective dimensions. Arrays require a subscript in each
dimension, in order to locate a given element of the array. Arrays are limited to six
dimensions, and the subscript range for each dimension must lie between -32767 and 32767.
REAL arrays require eight bytes of memory for each element, plus overhead. It is easy to see
that large arrays can demand massive memory resources.

An undeclared array is given as many dimensions as it has subscripts in its lowest-numbered
occurrence. Each dimension of an undeclared array has an upper bound of ten. Space for
these elements is reserved whether you use them or not.

Dimensioning an Array

Before you use an array, you should tell the system how much memory to reserve for it. This
is called “dimensioning” an array. You can dimension arrays with the DIM, COM, ALLOCATE,
INTEGER or REAL statements. For example:

REAL Array_complex(2,4)

An array is a type of variable and as such follows all rules for variable names. Unless you
explicitly specify INTEGER type in the dimensioning statement, arrays default to REAL type.
The same array can only be dimensioned once in a context (there is one exception to this
rule: If you ALLOCATE an array, and then DEALLOCATE it, you can dimension the array again).
However, as we explain later in this section, arrays can be REDIMensioned.

When you dimension an array, the system reserves space in internal memory for it. The
system also sets up a table which it uses to locate each element in the array. The location
of each element is designated by a unique combination of subscripts, one subscript for each
dimension. For a two-dimensional array, for instance, each element is identified by two
subscript values. An example of dimensioning a two-dimensional array is as follows:

OPTION BASE 0 default numbering of subscripts begins with 0
DIM Array(3,5)  declares elements (0,0) to (3,5)

OPTION BASE 1 default numbering of subscripts begins with 1
Array(2,3) defines elements (1,1) to (2,3)

OPTION BASE 0 default numbering of subscripts begins with 0
DIM A(1:4,1:4,1:4)  explicitly defines elements (1,1,1) to ({,4,4)

Each context defaults to an option base of 0 (but arrays appearing in COM statements with
an (*) keep their original base. However, you can set the option base to 1 using the OPTION
BASE statement. You can have only one OPTION BASE statement in a context, and it must
precede all explicit variable declarations.

Some Examples of Arrays

When we discuss two-dimensional arrays, the first dimension will always represent rows, and
the second dimension will always represent columns. Note also in the above example that the
first two dimensions use the default setting of 1 for the lower bound, while the third dimension
explicitly defines 0 as the lower bound. The numbers in parentheses are the subscript values
for the particular elements. These are the numbers you use to identify each array element.

The following examples illustrate some of the flexibility you have in dimensioning arrays.

10 OPTION BASE 1
20 DIM A(3,4,0:2)

oo 210 G [N
s \\\\\\\\ o
\ \,_

1st DIMENSION
Planes of a Three-Dimensional REAL Array

Dimension Size Lower Bound Upper Bound
st 3 1 3
2nd 4 1 4
3rd 3 0 2

10  OPTION BASE 1
20 COM B(1:5,2:6)

10
20

Two-Dimensional REAL ARRAY

(1,2) (1,3) (1,4) (1,5) (1,6)
(2,2) (2,3) (2,4) (2,5) (2,6)
(3,2) (3,3) (3,4) (3,5) (3,6)
(4,2) (4,3) (4,4) (4,5) (4,6)
(5,2) (5,3) (5,4) (5,5) (5,6)
Dimension Size Lower Bound Upper Bound
1st 5 1 5
2nd 5 2 6

OPTION BASE 1
ALLOCATE INTEGER C(2:4,-2:2)

A Dynamically Allocated, Two-Dimensional INTEGER Array

(2v'2) (2)'1) (2’0) (2’1) (2’2)
(3:‘2) (3)'1) (3v0) (3»1) (3)2)
(4-2) (4-1) (4,0) (4,1) (4.2)
Dimension Size Lower Bound Upper Bound
1st 3 2 4
2nd 5 -2 2

Note Throughout this chapter we will be using DIM statements without specifying
i what the current option base setting is. Unless explicitly specified otherwise,
% all examples in this chapter use option base 1.

As an example of a four-dimensional array, consider a five-story library. On each floor there
are 20 stacks, each stack contains 10 shelves, and each shelf holds 100 books. To specify the
location of a particular book you would give the number of the floor, the stack, the shelf,
and the particular book on that shelf. We could dimension an array for the library with the
statement:

DIM Library(5,20,10,100)

This means that there are 100,000 book locations. To identify a particular book you would
specify its subscripts. For instance, Library(2,12,3,35) would identify the 35th book on the
3rd shelf of the 12th stack on the 2nd floor.

Problems with Implicit Dimensioning

In any context, an array must have a dimensioned size. It may be explicitly dimensioned
through COM, INTEGER, REAL, or ALLOCATE. It can also be implicitly dimensioned through a
subscripted reference to it in a program statement other than a MAT or a REDIM statement.
MAT and REDIM statements cannot be used to implicitly dimension an array.

Finding Out the Dimensions of an Array

There are a number of statements that allow you to determine the size of an array. To find
out how many dimensions are in an array, use the RANK function. For example, this program

10 OPTION BASE 0
20 DIM F(1,4,-1:2)
30 PRINT RANK (F)
40 END

would print 3.

The SIZE function returns the size (number of elements) of a particular dimension. For
instance,

SIZE (F,2)
would return 5, the number of elements in F’s second dimension.

To find out what the lower bound of a dimension is, use the BASE function. Referring again to
array F,

BASE (F,1)

would return a 0, while,

BASE (F,3)
would return a -1, indicating this dimension has not been defined as part of F.

By using the SIZE and BASE functions together, you can determine the upper bounds of any
dimension (e.g., SIZE+BASE-1=Upper Bound).

These functions are powerful tools for writing programs that perform functions on an array
regardless of the array’s size or shape.

Using Individual Array Elements

This section deals with assigning and extracting individual elements from arrays.

Assigning an Individual Array Element

Initially, every element in an array equals zero. There are a number of different ways to
change these values. The most obvious is to assign a particular value to each element. This is
done by specifying the element’s subscripts.

A(3,4)=13  the element in row 8, column 4, has the value 13

Extracting Single Values From Arrays

As with entering values into arrays, there are a number of ways to extract values as well. To
extract the value of a particular element, simply specify the element’s subscripts.

X=A(3,4,2)

BASIC automatically converts variable types. For example, if you assign an element from a
REAL array to an INTEGER variable, the system will round the REAL to an integer.

Filling Arrays

This section discusses three methods for filling an entire array:
m Assigning every element the same value

m Using READ to fill an entire array

m Copying arrays into other arrays

Assigning Every Element in an Array the Same Value

For some applications, you may want to initialize every element in an array to some particular
value. You can do this by assigning a value to the array name. However, you must precede
the assignment with the MAT keyword.

MAT A= (10)

Note that the numeric expression on the right-hand side of the assignment must be enclosed in
parentheses and that this expression may be INTEGER or REAL.

Using the READ Statement to Fill an Entire Array

You can assign values to an array using READ and DATA. DATA allows you to create a stream of
data items, and READ enables you to enter the data stream into an array.

110  DIM A(3,3)

120  DATA -4,36,2.3,5,89,17,-6,-12,42

130  READ A(¥)

140  PRINT USING "3(3DD.DD,3DD.DD,3DD.DD,/)";A(*)
150 END

The asterisk in line 140 is used to designate the entire array rather than a single element.
Note also that the right-most subscript varies fastest. In this case, it means that the system
fills an entire row before going to the next one. The READ/DATA statements are discussed
further in the chapter “Data Storage and Retrieval”.

Executing the previous program produces the following results:

-4.00 36.00 2.30
5.00 89.00 17.00
-6.00 -12.00 42.00

Copying Entire Arrays into Other Arrays

Another way to fill an array is to copy all elements from one array into another (copying
sub-sets of arrays is discussed in the subsequent section of “Numeric Arrays” called “Copying
Subarrays”). Suppose, for example, that you have the two arrays A and B shown below.

0 0 0 3 5
A=10 0 0}|B=18 2
0 0 0 17

Note that A is a 3x3 array which is filled entirely with 0’s, while B is a 3X2 array filled with
non-zero values. To copy B to A, we would execute:

MAT A= B

Again, you must precede the assignment with MAT. The system will automatically redimension
the “result array” (the one on the left-hand side of the assignment) so that it is the same size
as the “operand array” (the one on the right side of the equation.) There are two restrictions
on redimensioning an array.

m The two arrays must have the same rank (e.g., the same number of dimensions.)

m The dimensioned size of the result array must be at least as large as the current size of the
operand array.

If BASIC cannot redimension the result array to the proper size, it returns an error.

Automatic redimensioning of an array will not affect the lower bounds, only the upper
bounds. So the BASE values of each dimension of the result array will remain the same. Also
keep in mind that the size restriction applies to the dimensioned size of the result array

and the current size of the operand array. Suppose we dimension arrays A, B and C to the
following sizes:

10  OPTION BASE 1
20 DIM A(3,3),B(2,2),0(2,4)

We can execute,

MAT A= B

since A is dimensioned to 9 elements and B is only 4 elements. The copy automatically
redimensions A to a 2X2 array. Nevertheless, we can still execute:

MAT A= C

This works because the nine elements originally reserved for A remain available until the
program is scratched. A now becomes a 2x4 matrix. After MAT A= C, we could not execute:

MAT B= A
or
MAT B= C

since in each of these cases, we are trying to copy a larger array into a smaller one. But we
could execute

MAT C= A

after the original MAT A= B assignment, since C’s dimensioned size (8) is larger than A’s current
size (4).

Printing Arrays

Printing an Entire Array

Certain operations (e.g., PRINT, OUTPUT, ENTER and READ) allow you to access all elements of
an array merely by using an asterisk in place of the subscript list. The statement,

PRINT A(*);
would display every element of A on the current PRINTER IS device. The elements are
displayed in order, with the rightmost subscripts varying fastest. The semi-colon at the end
of the statement is equivalent to putting a semi-colon between each element. When they are

displayed, therefore, they will be separated by a space. (The default is to place elements in
successive columns.)

Examples of Formatting Arrays for Display

This section provides two subprograms which have both been given the name Printmat.
The first subprogram is used to display a two-dimensional INTEGER array and the second
subprogram is used to display a three-dimensional INTEGER array.

To display a two dimensional array, you can use the following subprogram:

240 SUB Printmat (INTEGER Array(*))
250 OPTION BASE 1
260 FOR Row=BASE(Array,1) TO SIZE(Array,1)+BASE(Array,1)-1

270 FOR Column=BASE(Array,2) TO SIZE(Array,2)+BASE(Array,2)-1
280 PRINT USING "DDDD,XX,#";Array(Row,Column)

290 NEXT Column

300 PRINT

310 NEXT Row

320 SUBEND

Assuming that you intended to display a 5x5 array, your results should look similar to this:

11 12 13 14 156
21 22 23 24 25
31 32 33 34 35
41 42 43 44 45
51 52 B3 54 55

If you were to expand the above subprogram to print three-dimensional INTEGER arrays,
your subprogram would be similar to the following:

250  SUB Printmat (INTEGER Array(*))

260 OPTION BASE 1

270 FOR Zplane=BASE(Array,3) TO SIZE(Array,3)+BASE(Array,3)-1
280 PRINT TAB(6),"Plane ";Zplane

290 PRINT

300 FOR Yplane=BASE(Array,2) TO SIZE(Array,2)+BASE(Array,2)-1
310 FOR Xplane=BASE(Array,1) TO SIZE(Array,1)+BASE(Array,1)-1
320 PRINT USING "DDDD,XX,#";Array(Zplane,Yplane,Xplane)
330 NEXT Xplane

340 PRINT

350 NEXT Yplane

360 PRINT

370 NEXT Zplane

380  SUBEND

If you had a three dimensional array with the following dimensions:
DIM Arrayi(3,3,3)

filled with all 3’s, the results from executing the above subprogram would be as follows:

Plane 1
3 3 3
3 3 3
3 3 3
Plane 2
3 3 3
3 3 3
3 3 3
Plane 3
3 3 3
3 3 3
3 3 3

Passing Entire Arrays

The asterisk is also used to pass an array as a parameter to a function or subprogram. For
instance, to pass an array A to the Printmat subprogram listed earlier, we would write:

Printmat (A(%))

Copying Subarrays
An earlier section discussed copying the contents of an entire array into another entire array.
MAT Array55= Array33

Each element of Array33 is copied into the corresponding element of Array55 which is
redimensioned if necessary.

Now suppose you would like to copy a portion of one array and place it in a special location
within another array. This process is called copying subarrays.

Array4x4 Array3x4

1M1 12 13 14
21 {29716 | 24
311 91 -5 | 34
41 4243 44

45 67 -8
<+ | 44{-9 1612
99 91 =519

Copying a Subarray into Another Subarray

Topics discussed in this section are:

m Subarray specifier

m Copying a subarray into an array

m Copying an array into a subarray

m Copying a subarray into a subarray

m Copying a portion of an array into itself
m Rules for copying subarrays

Dimensions for the arrays covered in the above topics will assume an option base of 1 (OPTION
BASE 1) unless stated differently.

Subarray Specifier

A subarray is a subset of an array (an array within an array). A subarray is indicated after
the array name as follows:

Array_name (subarray_specifier)

String_array$ (subarray_specifier)

The above subarray could take on many “sizes” and “shapes” depending on what you used as
dimensions for the array and the values used in the subarray_specifier. Note that “size” refers

to the number of elements in the subarray and “shape” refers to the number of dimensions
and elements in each dimension, respectively [e.g. both of these subscript specifiers have
the same shape: (-2:1,-1:10) and (1:4,9:20)]. Before looking at ways you can express a
subarray, let’s learn a few terms related to the subarray specifier.

subscript range

is used to specify a set of elements starting with a beginning
element position and ending with a final element position. For
example, 5:8 represents a range of four elements starting with
element 5 and ending at element 8.

subscript expression is an expression which reduces the RANK of the subarray. For

default range

example if you wanted to select a one-element subarray from a
two-dimensional array which is located in the 2nd row and 3rd
column, you would use the following subarray specifier: (2,3:3).
The subscript expression in this subarray specifier is 2 which
restricts the subarray to row 2 of the array.

is denoted by an asterisk (i.e. (1,*)) and represents all of the
elements in a dimension from the dimension’s lower bound to
its upper bound. For example, suppose you wanted to copy the
entire first column of a two dimensional array, you would use
the following subarray specifier: (*,1), where * represents all
the rows in the array and 1 represents only the first column.

Some examples of subarray specifiers are as follows:

(1,%)

(1:2)

(*,-1:2)

(3,1:2)

(1’*’*)

(1,1:2,%)

(1’2)*)

(1:2,3:4)

a subscript expression and a default range which designate the first row of a
two-dimensional array.

a given subscript range which represents the first two elements of a
one-dimensional array.

a default range and subscript range which represents all of the elements in
the first four columns of a two-dimensional array (base of 2nd dimension
assumed to be -1).

a subscript expression and subscript range which represent the first two
elements in the third row of a two-dimensional array.

a subscript expression and two default ranges which represent a plane
consisting of all the rows and columns of the first plane in the first-dimension.

a subscript expression, subscript range and default range which represent the
first two rows in the first plane of the first-dimension.

two subscript expressions and a default range which represent the entire
second row in the first plane of the first-dimension.

two subscript ranges which represent elements located in the third and fourth
columns of the first and second rows of a two-dimensional array.

For more information on string arrays, see the “String Manipulation” chapter found in this

manual.

Copying an Array into a Subarray

In order to copy a source array into a subarray of a destination array, the destination array’s
subarray must have the same size and shape as the source array.

A destination and source array are dimensioned as follows:

100 OPTION BASE 1
110 DIM Des_array(-3:1,5),Sor_array(2,3)
Suppose these arrays contain the following INTEGER values:

Des_array Sor_array

1" 12 13 14 15
21 22 23 24 25 TTTTTSTTS
31 32 33 34 35 21 29 23

41 42 43 44 45
51 52 53 54 55

you would copy the source array Sor_array into a subarray of the destination array
Des_array by using program line 190 given below:

190 MAT Des_array(-1:0,2:4)= Sor_array

Des_array would have the following values in it as the result of executing the above
statement:

Des_array

1M1 12 13 14 15
21 22 23 24 25
31 111 12 13f 35
41 {21 22 23| 45
51 52 53 54 55

Copying a Subarray into an Array

A subarray can be copied into an array as long as the array can be re-dimensioned to be the
size and shape of the subarray specifier.

A destination and source array are dimensioned as follows:

100 OPTION BASE 1
110 DIM Des_array(8),Sor_array(-5:4)

Suppose both of these one-dimensional arrays contain the following values:

Des_array Sor_array

(-1 14 8 4 98 43 90 -3)(-11i-4 1 2 3 4 78 {100 8 18)

you would copy a subarray of the source array (Sor_array) into a destination array
(Des_array) by using program line 190 given below:

"190 MAT Des_array= Sor_array(-4:1)

Des_array will be re-dimensioned to have six elements with the following values in it as a
result of executing the above statement.

Des_array
((+ 1+ 2 3 4 7d)

Copying a Subarray into Another Subarray

Subarray specifiers must have the same size and shape when you are copying one subarray
into another.

A destination and source array are dimensioned as follows:

100 OPTION BASE 1
110  DIM Des_array(3,2,2),Sor_array(2,3,2)
120

130

Suppose these three-dimensional arrays contain the following values:

Des_array 3] 3
212||3]|3
1Liff2]2

Sor_array [211]212

N

1/1«‘( 12 23125/2

15 et

1311132

W

in order to properly copy a source subarray (Sor_array(*,2,*)) into a destination subarray
using asterisks to represent the ranges of dimensions, you would use line 190 given below:

190  MAT Des_array(3,*,*)= Sor_array(*,2,%)

A three dimensional array with the following values in it would be the result of executing the
above statement.

Des_array

e
) £ e

Copying a Portion of an Array into Itself

If you are going to copy a subarray of an array into another portion of the same array, the
two subarray locations should not overlap (e.g., MAT Array(2:4,1:3)= Array(1:3,2:4) is
an improper assignment). No error message will result from this misuse, but the result is
undefined.

A destination and source array are dimensioned as follows:

100  OPTION BASE 1
110  DIM Array(4,4)

Suppose this two dimensional array contains the following values:

Array

112 [13 14
21 22 |23 24
31 32 33 34
41 42 43 44

to copy a slice of this array into another portion of the same array, you would use program
line 190 given below:

190  MAT Array(3:4,1:2)= Array(1:2,3:4)

Array will have the following values in it as a result of executing the above statement.

11 12 13 14
21 22 23 24
13 14 33 34
23 24 43 44

Note that you cannot copy a subarray into the array it is part of with an implied
re-dimensioning of the array. A statement of the form:

MAT Array= Array(subarray_specifier)

will always generate a run-time error.

Rules for Copying Subarrays

This section should help limit the number of syntax and runtime errors you could make when
copying subarrays. A previous section titled “Subarray Specifier” provided you with examples
of the correct way of writing subarray specifiers for copying subarrays. In this section, you will
be given rules to things you should not do when copying subarrays. The rules are as follows:

m Subarray specifiers must not contain all subscript expressions (i.e. (1,2,3) is not allowed, it
will produce a syntax error). This rule applies to all subscript specifiers.

m Subarray specifiers must not contain all asterisks (*) or default ranges (i.e. (*,*,*) is not
allowed, it will produce a syntax error). This rule applies to all subscript specifiers.

m If two subarrays are given in a MAT statement, there must be the same number of ranges in
each subarray specifier. For example:

MAT Des_array1(1:10,2:3)= Sor_array(5:14,*,3)

is the correct way of copying a subarray into another subarray provided the default range
given in the source array (Sor_array) has only two elements in it. Note that the source
array is a three-dimensional array. However, it still meets the criteria of having the same
number of ranges as the destination array because two of its entries are ranges and one is an
expression.

m If two subarrays are given in a MAT statement, the subscript ranges in the source array
must be the same shape as the subscript ranges in the destination array. For example, the
following example is legal:

MAT Des_array(1:5,0:1)= Sor_array(3,1:5,6:7)
however, the following example is not legal:
MAT Des_array(0:1,1:5)= Sor_array(1:5,0:1)

because both of its subarray specifiers do not have the same shape (i.e. the rows and
columns in the destination array do not match the rows and columns in the source array).

Redimensioning Arrays

In our discussion of copying arrays we saw that the system automatically redimensions an
array if necessary. BASIC also allows you to explicitly redimension an array with the REDIM
statement. As with automatic redimensioning, the following two rules apply to all REDIM
statements:

m A REDIMed array must maintain the same number of dimensions.

®m You cannot REDIM an array so that it contains more elements than it was originally
dimensioned to hold.

Suppose A is the 3x3 array shown below.
1 2 3
A=14 5 6
7 89

To redimension it to a 2x4 array, you would execute:

REDIM A(2,4)

The new array now looks like the figure below:

1 2 3 4
A_[5678]

Note that it retains the values of the elements, though not necessarily in the same locations.
For instance, A(2,1) in the original array was 4, whereas in the redimensioned array it equals
5. For example, if we REDIMed A again, this time to a 2x2 array, we would get:

(3

We could then initialize all elements to 0:

REDIM A(0:1,0:1)

MAT A= (0)

0 0
A= 0 0 )
It is also important to realize that elements that are out of range in the REDIMed array still
retain their values. The fifth through ninth elements in A still equal 5 through 9 even though
they are now inaccessible. If we REDIM A back to a 3x3 array, these values will reappear. For
example:

REDIM A(3,3)

results in:

0 0 0
A=10 5 6
7 8 9

One of the major strengths of the REDIM statement is that it allows you to use variables for
the subscript ranges: this is not allowed when you originally dimension an array. In effect,
this enables you to dynamically dimension arrays. This should not be confused with the
ALLOCATE statement which allows you to dynamically reserve memory for arrays. In the
example below, for instance, we enter the dimensions from the keyboard.

10  OPTION BASE 1

20  INTEGER A(100,100)

30 INPUT "Enter lower and upper bounds of dimensions',
Low1,Up1,Low2,Up2

40 IF (Upil-Lowi1+1)*(Up2-Low2+1)>10000 THEN Too_big

50  REDIM A(Low1:Up1,Low2:Up2)

Line 40 tests to see whether the new dimensions are too big. If so, program control is passed
to a line labelled “Too_big”. If line 40 were not present, the REDIM statement would return an
error if the dimensions were too large.

String Manipulation

It is often desirable to store non-numerical information in the computer. A word, a name or a
message can be stored in the computer as a string. Any sequence of characters may be used in
a string. Quotation marks delimit the beginning and ending of the string. The following are
valid string assignments:

LET A$="COMPUTER"

Fail$="The test has failed."

File_name$="INVENTORY"
Test$=Fail$[5,8]

The left-hand side of the assignment (the variable name) is equated to the right-hand side of
the assignment (the literal). String variable names are identical to numeric variable names
with the exception of a dollar sign ($) appended to the end of the name.

The length of a string is the number of characters in the string. In the previous example, the
length of A$ is 8 since there are eight characters in the literal “COMPUTER”. A string with
length O (i.e., that contains no characters) is known as a null string.

HP Instrument BASIC allows the dimensioned length of a string to range from 1 to 32 767
characters. The current length (number of characters in the string) ranges from zero to the
dimensioned length.

The default dimensioned length of a string is 18 characters. The DIM and COM statements
define string lengths up to the maximum length of 32 767 characters. An error results
whenever a string variable is assigned more characters than its dimensioned length.

A string may contain any character. The only special case is when a quotation mark needs to
be in a string. Two quotes, in succession, will embed a quote within a string.

10 Quote$="The time is ""NOW""."
20 PRINT Quote$
30 END

produces

The time is "NOW".

String Storage

Strings with a length that exceed the default length of 18 characters must have space reserved
before assignment. The following statements may be used:

DIM Long$ [400] Reserve space for a 400 character string.
COM Line$[80] Reserve an 80 character common variable.
ALLOCATE Search$[Length] Dynamic length allocation.

The DIM statement reserves storage for strings.
DIM Part_number$[10] ,Description$[64],Cost$[5]

The COM statement defines common variables that can be used by subprograms.
COM Name$ [40] ,Phone$ [14]

Strings that have been dimensioned but not assigned return the null string.

String Arrays
Large amounts of text are easily handled in arrays. For example,
DIM File$(1:1000) [80]

reserves storage for 1000 lines of 80 characters per line. Do not confuse the brackets, which
define the length of the string, with the parentheses, which define the number of strings in the
array. Each string in the array can be accessed by an index. For example,

PRINT File$(27)

prints the 27th element in the array. Since each character in a string uses one byte of memory
and each string in the array requires as many bytes as the length of the string, string arrays
can quickly use a lot of memory.

A program saved on a disc as an ASCII type file can be entered into a string array,
manipulated, and written back out to disc.

Evaluating Expressions Containing Strings
This section covers the following topics:

m Evaluation Hierarchy

m String Concatenation

m Relational Operations

Evaluation Hierarchy

Evaluation of string expressions is simpler than evaluation of numerical expressions. The
three allowed operations are extracting a substring, concatenation, and parenthesization. The
evaluation hierarchy is presented in the following table.

Order Operation

High Parentheses

— Substrings and Functions

Low Concatenation

String Concatenation

Two separate strings are joined together by using the concatenation operator “&”. The
following program combines two strings into one:

10 One$="WRIST"

20 Two$="WATCH"

30 Concat$=0One$&Two$

40 PRINT One$,Two$,Concat$
50 END

prints
WRIST WATCH WRISTWATCH

The concatenation operation, in line 30, appends the second string to the end of the first
string. The result is assigned to a third string. An error results if the concatenation operation
produces a string that is longer than the dimensioned length of the string being assigned.

Relational Operations

Most of the relational operators used for numeric expression evaluation can also be used for
the evaluation of strings.

The following examples show some of the possible tests:

"ABC" = "ABC" True
"ABC" = " ABC" False
"ABC'" < "AbC" True
MM > N7 False
AVARS VA False
"long" <= "longer" True
"RE-SAVE" >= "RESAVE" False

Any of these relational operators may be used: <, >, <=, >=, =, <>.

Testing begins with the first character in the string and proceeds, character by character, until
the relationship has been determined.

The outcome of a relational test is based on the characters in the strings not on the length of
the strings. For example,

"BRONTOSAURUS" < "CAT"
is a true relationship since the letter “C” is higher in ASCII value than the letter “B”.

Substrings

You can append a subscript to a string variable name to define a substring. A substring may
comprise all or just part of the original string. Brackets enclose the subscript which can be a
constant, variable, or numeric expression. For example,

String$[4]

specifies a substring starting with the fourth character of the original string. The subscript
must be in the range 1 to the current length of the string plus 1. Note that the brackets
now indicate the substring’s starting position instead of the total length of the string as
when reserving storage for a string. Subscripted strings may appear on either side of the
assignment.

Single-Subscript Substrings

When a substring is specified with only one numerical expression, enclosed with brackets, the
expression is evaluated and rounded to an integer indicating the position of the first character
of the substring within the string.

The following examples use the variable A$, which has been assigned the literal
“DICTIONARY”:

Statement Output
PRINT A$ DICTIONARY
PRINT A$[0] (error)
PRINT A$[1] DICTIONARY
PRINT A$[5] IONARY
PRINT A$[10] Y

PRINT A$[11] (null string)
PRINT A$[12] (error)

When you use a single subscript it specifies the starting character position, within the string,
of the substring. An error results when the subscript evaluates to zero or greater than the
current length of the string plus 1. A subscript that evaluates to 1 plus the length of the
string returns the null string ("") but does not produce an error.

Double-Subscript Substrings

A substring may have two subscripts, within brackets, to specify a range of characters. When
a comma is used to separate the items within brackets, the first subscript marks the beginning
position of the substring, while the second subscript is the ending position of the substring.
The form is: A$[Start,End]. For example, if A$ = “JABBERWOCKY?”, then

A$[4,6] specifies the substring BER

When a semicolon is used in place of a comma, the first subscript again marks the beginning
position of the substring, while the second subscript is now the length of the substring. The
form is: A$[Start;Length].

A$[4;6] specifies the substring BERWOC

In the following examples, the variable B$ has been assigned the literal
“ENLIGHTENMENT”:

Statement Output

PRINT B$ ENLIGHTENMENT
PRINT B$[1,13] ENLIGHTENMENT
PRINT B$[1;13] |ENLIGHTENMENT
PRINT B$[1,9] ENLIGHTEN
PRINT B$[1;9] ENLIGHTEN
PRINT B$[3,7] LIGHT

PRINT B$[3;7] LIGHTEN

PRINT B$[13,13] |N
PRINT B$[13;1] N
PRINT B$[13,26] |[(error)
PRINT B$[13;13] |(error)
PRINT B$[14;1] (null string)

An error results if the second subscript in a comma separated pair is greater than the current
string length plus 1 or if the sum of the subscripts in a semicolon separated pair is greater
than the current string length plus 1.

Specifying the position just past the end of a string returns the null string.

Special Considerations

All substring operations allow a subscript to specify the first position past the end of a string.
This allows strings to be concatenated without the concatenation operator. For example,
10 A$="CONCAT"

20 A$[7]1="ENATION"
30 PRINT A$

40 END
produces
CONCATENATION

The substring assignment is only valid if the substring already has characters up to the
specified position. Access beyond the first position past the end of a string results in the error

ERROR 18 String ovfl. or substring err

It’s good practice to dimension all strings including those shorter than the default length of
eighteen characters.

String-Related Functions

Several intrinsic functions are available in HP Instrument BASIC for the manipulation of
strings. These functions include conversions between string and numeric values.

Current String Length

The “length” of a string is the number of characters in the string. The LEN function returns
an integer with a value equal to the string length. The range is from 0 (null string) through
32 767. For example,

PRINT LEN("HELP ME")
prints

7

Substring Position

The “position” of a substring within a string is determined by the POS function. The
function returns the value of the starting position of the substring or zero if the entire
substring was not found. For example,

PRINT POS("DISAPPEARANCE","APPEAR")
prints
4

Note that POS returns the first occurrence of a substring within a string. By adding a
subscript and indexing through the string, the POS function can be used to find all occurrences
of a substring.

String-to-Numeric Conversion

The VAL function converts a string expression into a numeric value. The number will be
converted to and from scientific notation when necessary. For example,

PRINT VAL("123.4E3")
prints
123400

The string must evaluate to a valid number or error 32 will result.

ERROR 32 String is not a valid number

The NUM function converts a single character into its equivalent numeric value. The number
returned is in the range: 0 to 255. For example,

PRINT NUM("A")
prints 65

Numeric-to-String Conversion

The VAL$ function converts the value of a numeric expression into a character string. The
string contains the same characters (digits) that appear when the numeric variable is printed.
For example,

PRINT 1000000,VAL$(1000000)
prints
1.E+6 1.E+6

The CHRS function converts a number into an ASCII character. The number can be of type
INTEGER or REAL since the value is rounded, and a modulo 255 is performed. For example,

PRINT CHR$(97) ;CHR$ (98) ; CHR$ (99)
prints

abc

String Functions

This section covers string functions, which perform the following tasks:
m reversing the characters in a string

m repeating a string a given number of times

m trimming the leading and trailing blanks in a string

m converting string characters to the desired case

String Reverse

The REVS$ function returns a string created by reversing the sequence of characters in the
given string. For example,

PRINT REV$ ("Snack cans")
prints

snac kcanS

String Repeat

The RPT$ function returns a string created by repeating the specified string, a given number
of times. For example,

PRINT RPT$("* *",10)
prints

X okk ckk okok dkk kk dkk dkk kok kk X

Trimming a String

The TRIMS$ function returns a string with all leading and trailing blanks (ASCII spaces)
removed. For example,

PRINT "*";TRIM(" 1.23  ");"x"
prints

*1.23%

Case Conversion

The case conversion functions, UPC$ and LWCS, return strings with all characters converted
to the proper case. UPC$ converts all lowercase characters to their corresponding uppercase
characters and LWC$ converts any uppercase characters to their corresponding lowercase
characters.

10 DIM Word$[160]

20 INPUT "Enter a few characters",Word$
30 PRINT

40 PRINT "You typed: ";Word$

50 PRINT "Uppercase: ";UPC$(Word$)

60 PRINT "Lowercase: '";LWC$(Word$)

70 END

Number-Base Conversion

Utility functions are available to simplify the calculations between different number bases.
The two functions IVAL and DVAL convert a binary, octal, decimal, or hexadecimal string
value into a decimal number. The IVAL$ and DVALS$ functions convert a decimal number
into a binary, octal, decimal, or hexadecimal string value. The IVAL and IVAL$ functions
are restricted to the range of INTEGER variables (-32 768 through 32 767). The DVAL
and DVALS functions allow “double length” integers and thus allow larger numbers to be
converted (-2 147 483 648 through 2 147 483 647).

Each function has two parameters: the number or string to be converted and the radix.
The radix is limited to the values 2, 8, 10 and 16, and represents the numeric base of the
conversion.

For example,

PRINT DVAL("FF5900",16)
PRINT IVAL("AA",16)
PRINT DVAL$(100,8)
PRINT IVAL$(-1,16)

prints

1.6734464E+7
170
00000000144
FFFF

Subprograms and
User-Defined Functions

One of the most powerful constructs available in any language is the subprogram. A
subprogram can do everything a main program can do except that it must be invoked or
“called” before it is executed, whereas a main program is executed by an operator. This
chapter describes the benefits of using subprograms and shows many of the details of using
them.

A user-defined function is simply a special form of subprogram.

Benefits of Subprograms

A subprogram has its own “context” or state that is distinct from a main program and all
other subprograms. This means that every subprogram has its own set of variables, its own
softkey definitions, its own DATA blocks, and its own line labels. There are several benefits to
be realized by taking advantage of subprograms:

m The subprogram allows the programmer to take advantage of the top-down design method
of programming.

m The program is much easier to read using the subprogram calls.

m By using subprograms and testing each one independently of the others, it is easier to locate
and fix problems.

m You may want to perform the same task from several different areas of your program.

m Libraries of commonly used subprograms can be constructed for widespread use.

A Closer Look at Subprograms

This section shows a few of the details of using subprograms.

Calling and Executing a Subprogram

A SUB subprogram is invoked explicitly using the CALL statement. A nuance of SUB
subprograms is that the CALL keyword is optional when invoking a SUB subprogram.

The omission of the CALL keyword when invoking a SUB subprogram is left solely to the
discretion of the programmer; some will find it more aesthetic to omit CALL, others will
prefer its inclusion. There are, however, two instances that require the use of CALL when
invoking a subprogram.

CALL is required
1. if the subprogram is called after the THEN keyword in an IF statement
2. in an ON..event..CALL statement

Differences Between Subprograms and Subroutines
A subroutine and a subprogram are very different in HP Instrument BASIC.

m The GOSUB statement transfers program execution to a subroutine. A subroutine is a
segment of program lines within the current context. No parameters need to be passed, since
it has access to all variables in the context (which is also the context in which the “calling”
segment exists).

m The CALL statement transfers program execution to a subprogram, which is in a separate
context. Subprograms can have pass parameters, and they can have their own set of local
variables that are separate from all variables in all other contexts.

Subprogram Location

A subprogram is located after the body of the main program, following the main program’s
END statement. (The END statement must be the last statement in the main program
except for comments.) Subprograms may not be nested within other subprograms, but are
physically delimited from each other with their heading statements (SUB or DEF) and ending
statements (SUBEND or FNEND).

Subprogram and User-Defined Function Names

A subprogram has a name that may be up to 15 characters long, just as with line labels and
variable names. Here are some legal subprogram names:

Initialize

Read_dwm

Sort_2_d_array

Plot_data

Because up to 15 characters are allowed for naming subprograms, it is easy and convenient
to name subprograms in such a way as to reflect the purpose for which the subprogram was
written.

Difference Between a User-Defined Function and a Subprogram

A SUB subprogram (as opposed to a function subprogram) is invoked explicitly using the
CALL statement. A function subprogram is called implicitly by using the function name in an
expression. It can be used in a numeric or string expression the same way a constant would be
used, or it can be invoked from the keyboard. A function’s purpose is to return a single value
(either a REAL number or a string).

There are several functions that are built into the HP Instrument BASIC language that can
be used to return values, such as SIN, SQR, EXP, etc.

Y=SIN(X)+Phase
Root1=(-B+SQR(B*B-4*A*C) )/ (2*A)

User defined functions can extend HP Instrument BASIC if you need a feature that is not
provided.

6-2 Subprograms and
User-Defined Functions



X=FNFactorial (N)
Angle=FNAtn2(Y,X)

Here is a general guideline for taking a set of data and analyzing it to generate a single value,
then implementing the subprogram as a function. On the other hand, if you actually want to
change the data itself, generate more than one value as a result of the subprogram, or perform

any I/O activity, it is better to use a SUB subprogram.

REAL Precision Functions and String Functions

A function is allowed to return either a REAL or a string value. Let’s examine one that
returns a string. There are two primary differences: the first is that a $ must be added to

the name of a function that is to return a string. This is used both in the definition of the
function (the DEF statement) and when the function is invoked. The second difference is that
the RETURN statement in the function returns a string instead of a number.

200 PRINT FNAscii_to_hex$(A$)
1550 DEF FNAscii_to_hex$(A$)
1560 ! Each ASCII byte consists of two hex
1670 ! digits; pretty formatting dictates that
1680 ! a space be inserted between every pair
1690 ! of hex digits. Thus, the output string
1600 ! will be three times as long as the input
1610 ! string.
1620 !
1630 ! upper four bits lower four bits
1640 ! UUUU LLLL UUUU LLLL
1650 ! shift 4 bits 0000 1111 mask (15)
1660 ! 0000 UUUU 0000 LLLL final
1670 !
1680 INTEGER I,Length,Hexupper,Hexlower
1690 Length=LEN(A$)
1695 Length=3*Length
1700 DIM Temp$[Length]
1710 FOR I=1 TO Length
1720 Hexupper=SHIFT(NUM(A$[I]),4)
1730 Hexlower=BINAND (NUM(A$[I],15)
1740 Temp$[3*I-2;1]=FNHex$ (Hexupper)
1750 Temp$[3*I-1;1]=FNHex$(Hexlower)
1760 Temp$[3*I;1]=" "
1770 NEXT I
1780 RETURN Temp$
1790 FNEND
1800 DEF FNHex$(INTEGER X)
1810 ! Assume 0<=X<=15)
1820 ! Return ASCII representation of the
1830 ! hex digit represented by the four
1840 ! bits of X.
1850 ! If X is between 0 and 9, return
1860 ! "o, .. "M
1870 ! If X > 9, return "A"..."F"
(Continued)

1880 IF X<=9 THEN

1890 RETURN CHR$(48+X) ! ASCII 48 through 57
1900 ! represent "Q'" - '"9"
1910 ELSE

1920 RETURN CHR$(55+X) ! ASCII 65 through 70
1930 ! represent "A'" - "F'"
1940 END IF

1950 FNEND

Lines 200, 1740, and 1750 show examples of how to call a string function. Lines 1550 and 1800
show where the two string function subprograms begin. Notice that the program could be
optimized slightly by deleting lines 1720 and 1730 and modifying lines 1740 and 1750:

1740 Temp$ [3*I-2; 1]1=FNHex$ (SHIFT(NUM(A$(I]) ,4))
1750 Temp$ [3*%I-1; 1]=FNHex$ (BINAND (NUM(A$[I],15))

Thus, it is perfectly legal to use expressions in the pass parameter list of a subprogram.

Program/Subprogram Communication

As mentioned earlier, there are two ways for a subprogram to communicate with the main
program or with other subprograms:

m By passing parameters

m By sharing blocks of common (COM) variables.

Parameter Lists
There are two places where parameter lists occur:
m The pass parameter list is in the CALL statement or FN call:
30 CALL Build_array(Numbers(*),20) ! Subprogram call.
50 PRINT FNSum_array(Numbers(*),20) ! User-defined function call.

It is known as the pass parameter list because it specifies what information is to be passed
to the subprogram.

m The formal parameter list is in the SUB or DEF FN statement that begins the
subprogram’s definition:

70 SUB Build_array(X(*),N) ! Subprogram "Build_array".
410 DEF FNSum_array(A(*),N) ! User-defined function "Sum_array".
This is known as the formal parameter list because it specifies the form of the information
that can be passed to the subprogram.
Formal Parameter Lists

The formal parameter list is part of the subprogram’s definition, just like the subprogram’s
name. The formal parameter list defines

m the number of values that may be passed to a subprogram

m the types of those values (string, INTEGER, or REAL, and whether they are simple or
array variables; or I/O path names)

m the variable names the subprogram will use to refer to those values. (This allows the name
in the subprogram to be different from the name used in the calling context.)

The subprogram has the power to demand that the calling context match the types declared
in the formal parameter list—otherwise, an error results.

Pass Parameter Lists

The calling context provides a pass parameter list that corresponds with the formal parameter
list provided by the subprogram. The pass parameter list provides

m the actual values for those inputs required by the subprogram.

m storage for any values to be returned by the subprogram (pass by reference parameters
only).

It is perfectly legal for both the formal and pass parameter lists to be null (non-existent).

Passing By Value vs. Passing By Reference
There are two ways for the calling context to pass values to a subprogram:
m pass by value—the calling context supplies a value and nothing more.

m pass by reference—the calling context actually gives the subprogram access to the calling
context’s value area (which is essentially access to the calling context’s variable).

The distinction between these two methods is that a subprogram cannot alter the value of
data in the calling context if the data is passed by value, while the subprogram can alter the
value of data in the calling context if the data is passed by reference.

The subprogram has no control over whether its parameters are passed by value or passed by
reference. That is determined by the calling context’s pass parameter list. For instance, in the
example below, the array Numbers(*) is passed by reference, while the quantity 20 is passed
by value.

30 CALL Build_array(Numbers(*),20) ! Subprogram call.
The general rules for passing parameters are as follows:

m In order for a parameter to be passed by reference, the pass parameter list (in the calling
context) must use a variable for that parameter.

m In order for a parameter to be passed by value, the pass parameter list must use an
expression for that parameter.

Note that enclosing a variable in parentheses is sufficient to create an expression and that
literals are expressions. Using pass by value, it is possible to pass an INTEGER expression
to a REAL formal parameter (the INTEGER is converted to its REAL representation)
without causing a type mismatch error. Likewise, it is possible to pass a REAL expression
to an INTEGER formal parameter (the value of the expression is rounded to the nearest
INTEGER) without causing a type mismatch error (an integer overflow error is generated if
the expression is out of range for an INTEGER).

Example Pass and Corresponding Formal Parameter Lists

Here is a sample formal parameter list showing which types each parameter demands:

SUB Read_dvm(@Dvm,A(*) ,INTEGER Lower,Upper,Status$,Errflag)

QDvm

A(x)

Lower, Upper

Status$

Errflag

This is an I/O path name that may refer to either an I/O device or a mass
storage file. Its name here implies that it is a voltmeter, but it is perfectly
legal to redirect I/O to a file just by using a different ASSIGN with @Dvm.

This is a REAL array. Its size is declared by the calling context. The
parameters Lower and Upper contain its limits.

These are declared here to be INTEGERs. Thus, when the calling program
invokes this subprogram, it must supply either INTEGER variables or
INTEGER expressions, or an error will occur.

This is a simple string that presumably could be used to return the status of
the voltmeter to the main program. The length of the string is defined by the
calling context.

This is a REAL number. The declaration of the string Status$ has limited
the scope of the INTEGER keyword which caused Lower and Upper to require
INTEGER pass parameters.

Let’s look at our previous example from the calling side (which shows the pass parameter list):

CALL Read_dvm(@Voltmeter,Readings(*),1,400,Status$,Errflag)

QVoltmeter

Readings(x)

1, 400

Status$

Errflag

This is the pass parameter that matches the formal parameter @Dvm in the
subprogram. I/O path names are always passed by reference, which means
the subprogram can close the I/O path or assign it to a different file or
device.

This matches the array A(*) in the subprogram’s formal parameter list.
Arrays, too, are always passed by reference.

These are the values passed to the formal parameters Lower and Upper. Since
constants are classified as expressions rather than variables, these parameters
have been passed by value. Thus, if the subprogram used either Lower or
Upper on the left-hand side of an assignment operator, no change would take
place in the calling context’s value area.

This is passed by reference here. If it was enclosed in parentheses, it would
be passed by value. Notice that if it was passed by value, it would be totally
useless as a method for returning the status of the voltmeter to the calling
context.

This is passed by reference.

COM Blocks

Since we’ve discussed parameter lists in detail, let’s turn now to the other method a
subprogram has of communicating with the main program or with other subprograms, the
COM block.

There are two types of COM (or common) blocks: blank and labeled. Blank COM is simply
a special case of labeled COM (it is the COM name that is nothing) with the exception that
blank COM must be declared in the main program, while labeled COM blocks don’t have to
be declared in the main program. Both types of COM blocks simply declare blocks of data
that are accessible to any context with matching COM declarations.

A blank COM block might look like this:
20 COM Conditions(15),INTEGER,Cmin,Cmax,@Nuclear_pile,Pile_status$[20], Tolerance

A labeled COM might look like this:
30 COM /Valve/ Main(10),Subvalves(10,15),@Valve_ctrl

A COM block’s name, if it has one, will immediately follow the COM keyword, and will be set
off with slashes, as shown above. The same rules used for naming variables and subprograms
are used for naming COM blocks.

Any context need only declare those COM blocks that it needs to have access to. If there are
150 variables declared in 10 COM blocks, it isn’t necessary for every context to declare the

